

```

// T31stEst.sce
// STATE ESTIMATION (KALMAN FILTER)
// Experiments
// - change model parameters M,N,A,B
// - set different system and model covariances rw,rv and Rw,Rv
// - try lower stat-estimate covariance Rx
// -----
exec("ScIntro.sce",-1), mode(0), getd()

nd=200;                                // number of data                // 1
// SIMULATION                          // 2
M=[.8 .1                               // parameters of simulation        // 3
   .3 .6];                             // 4
N=[.5 -.5]';                           // 5
A=[.9 -.2];                             // 6
B=0;                                    // 7
rw=.1*eye(2,2);                         // noise covariances              // 8
rv=.1;                                  // 9
x(:,1)=[0 0]';                          // initial state                  // 10
ut=signal(nd,1);                         // input                          // 11
// time loop of simulation              // 12
for t=2:nd                              // 13
    x(:,t)=M*x(:,t-1)+N*ut(t)+rw*rand(2,1,'n'); // 14

```

```

    y(t) =A*x(:,t)+B*ut(t)+rv*rand(1,1,'n');           // 15
end                                                     // 16
                                                       // 17
// ESTIMATION                                           // 18
// initialization of estimation                         // 19
Rw=.1*eye(2,2);                                       // state noise covariance // 20
Rv=.1;                                                // output noise covariance // 21
Rx=1000*eye(2,2);                                    // estimated state covariance // 22
xp(:,1)=zeros(2,1);                                  // initial state           // 23
// loop for state estimation                           // 24
for t=2:nd                                           // 25
    [xp(:,t),Rx,yp(t)]=Kalman(xp(:,t-1),y(t),ut(t),M,N,[],A,B,[],Rw,Rv,Rx); // 26
end                                                  // 27
                                                       // 28
// RESULTS                                             // 29
subplot(311),plot(1:nd,x(1,:),1:nd,xp(1,:))         // 30
set(gcf(),"position",[700 100 600 500])             // 31
title('First state entry')                           // 32
legend('state','estimate');                           // 33
subplot(312),plot(1:nd,x(2,:),1:nd,xp(2,:))         // 34
title('Second state entry')                           // 35
legend('state','estimate');                           // 36
subplot(313),plot(1:nd,y,1:nd,yp')                  // 37

```

```
title('Output') // 38
legend('output','estimate'); // 39
```

## Description of the program

The time loop of the program runs only the procedure `Kalman()`. However, it is necessary to do some preparations. Especially important and also difficult is setting the noise covariances of the model state and output  $r_w$  and  $r_w$ . They should be guessed rather precisely and there is no general way how to do it. They should reflect errors between prediction of the state and output and the real state and output. Problem is, that the real state is not measured.

Parameters  $M, N, A, B$  of the model are expected known (which also is not very practical).

- Rows 3–9 define parameters for simulation
- Row 10 sets the initial point estimate of the state for simulation
- Row 11 generates input signal
- Rows 14–15 perform simulation
- Rows 20–21 set covariances of the state and output. Here, the task is simple as we know what we have simulated.
- Row 23 defines initial point estimate of the state
- Row 26 performs in the time loop estimation of the state