

```

// T48mixDesPoi.sce
// MIXTURE ESTIMATION (descriptive, Poisson)
// Experiments
// - change simulated parameters
// - change initial parameters
// -----
exec("ScIntro.sce",-1),
getd(), mode(0)

nd=500; // 1
ni=1; // weight for initial data // 2
// PARAMETERS // 3
lamS=[.2 5 29]; // component parametrs // 4
nc=length(lamS); // number of components // 5
alS=[.3 .3 .4]; // parameters of pointer model // 6
// 7
// SIMULATION // 8
for t=1:nd // 9
    jS=sampCat(alS); // pointer value // 10
    cS(t)=jS; // stor pointer value // 11
    y(t)=sampPoi(lamS(jS)); // output generation // 12
end // 13
// 14

```

```

// ESTIMATION // 15
// initialization // 16
ka=[1 1 1]*ni; // initial counter // 17
lamE=[1 10 20]; // initial parameters // 18
S=lamE.*ka; // statistics for given pE // 19
// 20
// time loop // 21
for t=1:nd // 22
    for j=1:nc // 23
        q(j)=poiss(y(t),lamE(j)); // proximity // 24
    end // 25
    w=q/sum(q); // weights // 26
    wt(:,t)=w; // remember weights // 27
    for j=1:nc // 28
        S(j)=S(j)+w(j)*y(t); // update of // 29
        ka(j)=ka(j)+w(j); // statistics // 30
        lamE(j)=S(j)/ka(j); // parameter estimates // 31
        lamEt(t,j)=lamE(j); // remember // 32
    end // 33
end // 34
// 35
// RESULTS // 36
tx=['b';'r';'g']; // 37

```

```

set(scf(1),'position',[600 10 600 800]) // evolution of par. est.      // 38
    title 'Evolution of the estimated parameters'                      // 39
for j=1:nc                                // 40
    subplot(3,1,j)                  // 41
    plot(lamEt,'-')                // 42
    xlabel('component '+string(j)) // 43
end                                  // 44
                                     // 45
disp 'The final parmeter estimates are' // 46
disp(lamE)                          // 47
                                     // 48
[nill,cp]=max(wt,'r');               // accuracy of classification // 49
disp 'Accuracy of classification'    // 50
ACC=acc(cS,cp)                      // 51

```

## Description of the program

- Row 2 sets the strength of prior information ( $n_i$  is as if the number of prior data from which the information has been gained).
- Rows 4–6 define parameters of the task.
- Rows 8–13 perform simulation - generation from Poisson distribution of individual components.
- Rows 17–19 prepare initialization of the task

- Row 19 computes the statistics to be in accord with the values of initial the parameters
- Rows 21–34 run the time loop.
  - Row 24 computes the proximities.
  - Rows 26–27 construct the component weights.
  - Rows 29–32 update the statistics and recompute the values of the point estimates of the parameters.