

```

// T23preReg1.sce
// NP-STEP POINT PREDICTION WITH REGRESSION MODEL (KNOWN PARAMETERS)
// Experiments
// Change: - np = number of steps of prediction
//          - r = noise variance
//          - th = model parameters
//          - u = input signal
// Shift of regression vector
// ps(t)= [u(t) y(t-1) u(t-1) y(t-2) u(t-2) 1]      at time t
//          |-----|
// ps(t+1)=[u(t+1) y(t) u(t) y(t-1) u(t-1) 1]      at time t+1
//          |-----|
// -----
exec("ScIntro.sce",-1), mode(0)

// PARAMETERS // 1
nd=100; // number of data // 2
np=5; // length of prediction (np>=1) // 3
// b0 a1 b1 a2 b2 k // 4
thS=[1 .4 -.3 -.5 .1 1]'; // regression coefficients // 5
rS=.02; // noise variance // 6
u=signal(nd,1); // input // 7
y(1)=1; y(2)=-1; // prior data // 8

```

```

// 9
// TIME LOOP // 10
for t=3:(nd-np) // time loop (on-line tasks) // 11
    // prediction // 12
    ps=[u(t) y(t-1) u(t-1) y(t-2) u(t-2) 1]'; // first reg. vec // 13
    yy=ps'*thS; // zero prediction for time = t // 14
    for j=1:np // predictions for t+1,...,t+np // 15
        ps=[u(t+j); yy; ps(1:$-3); 1]; // reg.vecs with predicted outputs // 16
        yy=ps'*thS; // new prediction (partial) // 17
    end // 18
    yp(t+np)=yy; // final prediction for time t+np // 19
// 20
// simulation // 21
ps=[u(t) y(t-1) u(t-1) y(t-2) u(t-2) 1]'; // reg. vector for sim. // 22
y(t)=ps'*thS+sqrt(rS)*rand(1,1,'n'); // output generation // 23
end // 24
// 25
// RESULTS // 26
s=(np+3):(nd-np); // 27
scf(1); // 28
plot(s,y(s),'.: ',s,yp(s),'rx') // 29
set(gca(),"data_bounds",[1 nd -3 5]) // 30
legend('output','prediction'); // 31

```

```

title(string(np)+'-steps ahead prediction')           // 32
                                                    // 33
RPE=variance(y(s)-yp(s))/variance(y)    // relative prediction error    // 34

```

## Description of the program

Row 3 defines number of steps of the prediction

Rows 5–6 prepare values of parameters

Row 7 defines the input variable

Row 8 sets initial values of y

Rows 11–24 perform the time loop (as we predict when y is not measured, yet, the sequence of the tasks is: prediction (estimate of  $y(t)$ ), simulation (measuring of  $y(t)$ )

13: construction of regression vector

14: first step of prediction (based on measured data)

15–18: loop for internal predictions (all unknown values of y are substituted by heir estimates from the previous steps)

16: construction of regression vector (predictions of y from previous steps are included).

Note: For the shift of the regression vector see the head of the program.

17: generation of the new prediction

19: the final prediction at the time  $t+np$

22: regression vector fo simulation

23: simulation of new  $y(t)$