

```

// T11simReg1.sce
// SECOND ORDER REGRESSION MODEL
// Experiments
// - change parameters of the model
// - change the input signal
// - try to increase the model order to 3
// -----
exec("ScIntro.sce",-1), mode(0)

// PARAMETERS OF THE SIMULATION // 1
nd=100; // length of data // 2
aS=[.4 .2]; // parameters at y // 3
bS=[1 .2 -.5]; // parameters at u // 4
kS=5; // constant (model absolute term) // 5
sS=.1; // noise std // 6
// 7
y(1)=1; y(2)=3; // initial conditions for output // 8
u=signal(nd,1,.1); // input // 9
thS=[aS bS kS]'; // vector of parameters // 10
V=zeros(7,7); // 11
ka=0; // 12
// 13
// TIME LOOP // 14

```

```

for t=3:nd // 15
    ps=[y(t-1) y(t-2) u(t) u(t-1) u(t-2) 1]'; // regression vector // 16
    y(t)=thS'*ps+sS*rand(1,1,'n'); // simulation // 17
    Ps=[y(t) ps']'; // 18
    V=V+Ps*Ps'; // update of statistics // 19
    ka=ka+1; // 20
end // 21
Vy=V(1,1); Vyp=V(2:$,1); Vp=V(2:$,2:$); // 22
thE=inv(Vp)*Vyp; // parameter estimate // 23
// 24
// RESULTS OF THE SIMULATION // 25
set(gcf(),"position",[700 100 600 500]) // 26
subplot(211),plot(1:nd,u),title('Input') // 27
subplot(212),plot(1:nd,y),title('Output') // 28
disp('The simulated and estimated parameters are',[thS,thE]) // 29

```

Description of the program

- Rows 3–6 define parameters of the model

$$y_t = aS(1)y_{t-1} + aS(2)y_{t-2} + bS(1)u_t + bS(2)u_{t-1} + bS(3)u_{t-2} + kS + e_t,$$

sS is standard deviation of the noise e_t

- Row 8 defines initial values of y
- Row 9 generates input signal (see function `signal()`)
- Rows 11–12 set initial statistics V and κ (zero prior information)
- Rows 15–21 perform the time loop
 - 16: construction of regression vector
 - 17: simulation
 - 18: extended regression vector
 - 19–20: update of statistics
- Row 22 performs division of information matrix
- Row 23 computes point estimates of the parameters